Jacob Bernoulli's *Reisbüchlein* an RDF-star-based Edition

Ammann, Nora Olivia
n.ammann@stud.unibas.ch
Universität Basel, Switzerland

Alassi, Sepideh
sepideh.alassi@unibas.ch
Universität Basel, Switzerland

Rosenthaler, Lukas
lukas.rosenthaler@unibas.ch
Universität Basel, Switzerland

RDF-star is the technology best suited for the creation of LOD-based editions of metadata-oriented documents such as travel journals. Respectively, SPARQL-star is a powerful tool to query and analyze knowledge graphs with respect to the stored metadata information for triples (Alassi, Rosenthaler 2022). We developed an RDF-star-based ontology for travel journals; as a prototype, we have used Jacob Bernoulli's travel journal Reisbüchlein. Bernoulli travelled in the years 1676 to 1683 across Europe; during his journeys, he not only engaged in typical activities such as sightseeing, meeting friends, and excursions but also worked as a private teacher and gave lectures. He meticulously noted details about his itineraries, accommodations, means of transportation, food consumption, and corresponding costs in this journal. The manuscript was transcribed by a typewriter in the 1950s at the University of Basel. This transcription was scanned, and text files were created through Transkribus.[1] For our digital edition, we have developed a research-based ontology, trip, trying to stay as close as possible to the content without losing generality. This ontology extends the ontology of the Bernoulli-Euler Online project, beol,[2] that is used to create a digital edition of manuscripts and correspondence of members of the Bernoulli dynasty and Leonhard Euler (Alassi 2020: Chapter 5). We use other existing ontologies like foaf[3] and dbo[4] by making sub-classes and sub-properties.[5] To store and query the data with the defined ontology, we have used GraphDB Free,[6] which supports RDF-star and SPARQL-star.

Most of the information in documents such as travel journals is accompanied by metadata information describing it, e.g., "Jacob Bernoulli stayed at Hotel Schlüssel" would have metadata information such as the cost and duration of stay which are statements about a statement. RDF-star allows for triples that represent metadata about another triple by directly using this other triple as its subject or object (Hartig 2017). Thus, using RDF-star, the metadata information can be easily added to the predicate trip:stayedAt. These triples representing the metadata information can then have further metadata statements, and so on. Thus, we can have multiple levels of nested RDF-star statements, which best represent the information and can be easily queried using SPARQL-star.

A journey might include several stops; if a stop included an overnight stay it would be a sub-journey (represented by the predicate trip:hasSubJourneyTo) otherwise it is a transit (represented by the predicate trip:hasTransitThrough). Bernoulli's journey from Basel to Geneva included several sub-journeys with overnight stays represented by the class trip:stay. Figure 1 shows part of the RDF-star graph representing one of the sojourns en route to Geneva. The persons are represented with beol:person class (subclass of foaf:person) with GND numbers[7] given with the predicate beol:hasIAFIdentifier. Each town mentioned in the text is represented as trip:settlement class (subclass of dbo:settlement) with a unique identifier geoname-ID.[8]



Figure 1: Sub-journey to Liestal.

The sub-journey to Liestal is metadata about the main journey to Geneva, which itself has metadata information about the type of stay, which itself has metadata about the accommodation used. The currency of the cost of the accommodation is then given as the 4th level of nested RDF-star triples (Figure 2).

```
<<<<<<<<<< :bernoulli_jacob trip:travelled :journey_geneva >>
    trip:hasSubJourneyTo :liestal >>
    trip:hasStay :stay_liestal >>
    trip:stayedAt :accommodation_liestal >>
    trip:hasCost 2.0 >>
    trip:hasCurrency :currency_francken .
```

Figure 2: RDF-star triple about cost of accommodation.

Currencies are modeled as hierarchical RDF lists with subunits in a generic form. Figure 3 shows the query for the cost of Bernoulli's stay at Liestal en route to Geneva. The intuitive hierarchy of the RDF-star model is reflected in the SPARQL-star query. By binding triples to variables, we can express complex query criteria in a concise way.

```
PREFIX trip: <http://www.knora.org/ontology/0801/trip#>
PREFIX knora-base: <http://www.knora.org/ontology/knora-base#>
SELECT ?startLocationName ?dateSubJourney ?accommodationName ?dateStay ?costAccommodation ?currencyCost
WHERE {
    ?jacob a beol:person .
    ?jacob beol:hasIAFIdentifier "(DE-588)118509950" .
    BIND (<< ?jacob trip:travelled ?journey >> AS ?mainJourney)
    ?journey trip:hasDestination ?endDestination .
    ?endDestination trip:hasName "Genf"@de .
    BIND (<< ?mainJourney trip:hasSubJourneyTo ?destinationSubJourney >> AS ?subJourney)
    ?destinationSubJourney trip:hasName "Liestal"@de .
    ?subJourney trip:hasStartLocation ?startLocation ;
                trip:hasDate ?dateSubJourney ;
                trip:hasStay ?stay .
    ?startLocation trip:hasName ?startLocationName .
    ?stay trip:hasDate ?dateStay .
    BIND (<< ?stay trip:stayedAt ?accommodation >> AS ?accommodationStay)
    ?accommodation trip:hasName ?accommodationName .
    BIND (<< ?accommodationStay trip:hasCost ?costAccommodation >> AS ?cost)
    ?cost trip:hasCurrency ?currency .
    ?currency knora-base:listNodeName ?currencyCost .
}
```

Figure 3: SPARQL-star query for cost of stay at Liestal.

	startLocationName	dateSubJourney	accommodationName	dateStay	costAccommodation	currencyCost
1	"Basel"@de	"GREGORIAN:1676-08-20"	"Schlüssel"@de	"GREGORIAN:1676-08-20:1676-08-21"	"2.0"^^xsd:decimal	"Franken"

Figure 4: Results of the query in Figure 3.

By adding the start location of a sub-journey through the RDF-star triple trip:hasStartLocation we can represent the order of sub-journeys that later would facilitate the visualization of journeys. Figure 5 shows the Basel-Geneva journey with a focus on sub-journeys. All stages of a journey can be queried as shown in Figure 6.



Figure 5: Sub-journeys of Basel-Geneva journey.

```
PREFIX beol: <http://www.knora.org/ontology/0801/beol#>
PREFIX trip: <http://www.knora.org/ontology/0801/trip#>
SELECT ?dateSubJourney ?originSubJourney ?destinationSubJourney
WHERE {
    ?jacob a beol:person .
    ?jacob beol:hasIAFIdentifier "(DE-588)118509950" .
    BIND (<< ?jacob trip:travelled ?journey >> AS ?mainJourney)
    ?journey trip:hasDestination ?endDestination .
    ?endDestination trip:hasName "Genf"@de .
    BIND (<< ?mainJourney trip:hasSubJourneyTo ?subJourneyTo >> AS ?subJourney)
    ?subJourneyTo trip:hasName ?destinationSubJourney .
    ?subJourney trip:hasStartLocation ?startLocation ;
                trip:hasDate ?dateSubJourney .
    ?startLocation trip:hasName ?originSubJourney .
}
```

Figure 6: Query for sub-journeys enroute Geneva.

	dateSubJourney	originSubJourney	destinationSubJourney
1	"GREGORIAN:1676-08-20"	"Basel"@de	"Liestal"@de
2	"GREGORIAN:1676-08-21"	"Liestal"@de	"Waldenburg"@de
3	"GREGORIAN:1676-08-22"	"Waldenburg"@de	"Solothurn"@de
4	"GREGORIAN:1676-08-23"	"Solothurn"@de	"Biel"@de
5	"GREGORIAN:1676-08-24"	"Biel"@de	"Avenches"@de
6	"GREGORIAN:1676-08-25"	"Avenches"@de	"Lausanne"@de
7	"GREGORIAN:1676-08-26"	"Lausanne"@de	"Coppet"@de
8	"GREGORIAN:1676-08-27"	"Coppet"@de	"Genf"@de

Figure 7: Results of the query in Figure 6.

Similarly, we have defined an RDF-star-based data model to represent the activities Bernoulli undertook in his journeys, transits, etc.

Representing the same information using standard RDF would have required reification and processing overhead due to the increased number of additional statements needed to identify the reference triple and would appear too verbose (Kasenchak et al. 2021). The SPARQL-star queries, despite the nested levels of RDF-star triples, have lower query time and are easier to compose compared to standard SPARQL.

Notes

1. https://readcoop.eu/transkribus/.
2. https://beol.staging.dasch.swiss/.
3. http://xmlns.com/foaf/0.1/.
4. https://www.dbpedia.org/resources/ontology/.
5. Classes and predicates from knora-base, the ontology of our API is also used regarding connections to transcriptions and facsimile. (https://ezproxy.library.und.edu/login?url=https://docs.dasch.swiss/2022.08.01/DSP-API/02-knora-ontologies/knora-base/).
6. https://graphdb.ontotext.com.
7. https://www.dnb.de/DE/Professionell/Standardisierung/GND/gnd_node.html.
8. http://www.geonames.org/.

Bibliography

Alassi, Sepideh (2020): “From the Mechanics of Jacob Bernoulli to Digital History of Science. Infrastructure, Tools, and Methods” Ph.D. thesis. University of Basel. https://edoc.unibas.ch/81737/1/alassi_PhDThesis.pdf [17.04.2023].

Alassi, Sepideh / **Rosenthaler, Lukas** (2022): “RDF-star-based Digital Edition of Travel Journals”, in: *Digital Humanities 2022: Responding to Asian Diversity*, Tokyo, Japan, July 2022: 416 https://dh2022.dhii.asia/dh2022bookofabsts.pdf [17.04.2023].

Hartig, Olaf (2017): “RDF* and SPARQL*: An Alternative Approach to Annotate Statements in RDF”, in: Alberto Mendelzon Workshop on Foundations of Data Management, Montevideo, Uruguay, June 2017: https://ceur-ws.org/Vol-1912/paper12.pdf [17.04.2023].

Kasenchak, Bob / **Lehnert , Aren** / **Loh , Gene** (2021): “Use Case: Ontologies and RDF-Star for Knowledge Management”, in: *The Semantic Web: ESWC 2021 Satellite Events*, Hersonissos, Greece, June 2021: 254–260 https://doi.org/10.1007/978-3-030-80418-3_38 [17.04.2023].